Catalogue of Spacetimes

Thomas Müller and Frank Grave
Visualisierungsinstitut der Universität Stuttgart (VISUS), Germany

Barriola-Vilenkin metric

The Barriola-Vilenkin monopole metric in spherical coordinates (t, r, θ, ϕ) is given by the line element

$$ds^2 = -c^2\, dt^2 + dr^2 + k^2 r^2 (d\theta^2 + \sin^2\theta\, d\phi^2)$$

where $k<1$ and c is the speed of light.

May 29, 2009

Coordinates and metric:

■ Clearing the values, setting the dimension and defining a list of coordinates

```
Clear[coord, metric, inversemetric, affine, t, r, θ, ϕ]
```

```
n := 4
```

```
coord := {t, r, θ, ϕ}
```

■ Metric and inverse Metric

The metric $g_{\mu\nu}$ is given by the list

```
metric := {{-c^2, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, k^2 r^2, 0}, {0, 0, 0, k^2 r ^2 Sin[θ]^2}}
```

and the inverse metric $g^{\mu\nu}$ follows from

```
inversemetric := Simplify[Inverse[metric]]
```

■ Christoffel symbols of the second kind

The Christoffel symbols of the second kind are defined as $\Gamma^{\mu}_{\nu\lambda} = \frac{1}{2} g^{\mu\rho}\left(g_{\rho\nu,\lambda} + g_{\rho\lambda,\nu} - g_{\nu\lambda,\rho}\right)$

```
affine := affine = Simplify[Table[(1 / 2)
      Sum[inversemetric[[μ, ρ]] (D[metric[[ρ, ν]], coord[[λ]]] + D[metric[[ρ, λ]], coord[[ν]]] -
          D[metric[[ν, λ]], coord[[μ]]]), {ρ, 1, n}], {ν, 1, n}, {λ, 1, n}, {μ, 1, n}]]
```

```
listaffine := Table[If[UnsameQ[affine[[ν, λ, μ]], 0],
   {Style[Subsuperscript[Γ, Row[{coord[[ν]], coord[[λ]]}], coord[[μ]]], 18],
    "=", Style[affine[[ν, λ, μ]], 14]}], {λ, 1, n}, {ν, 1, λ}, {μ, 1, n}]
```

```
TableForm[Partition[DeleteCases[Flatten[listaffine], Null], 3], TableSpacing → {1, 2}]
```

$\Gamma^{\theta}_{r\theta}$	=	$\frac{1}{r}$
$\Gamma^{r}_{\theta\theta}$	=	$-k^2\, r$
$\Gamma^{\phi}_{r\phi}$	=	$\frac{1}{r}$
$\Gamma^{\phi}_{\theta\phi}$	=	Cot[θ]
$\Gamma^{r}_{\phi\phi}$	=	$-k^2\, r\, \mathrm{Sin}[\theta]^2$
$\Gamma^{\theta}_{\phi\phi}$	=	$-\mathrm{Cos}[\theta]\, \mathrm{Sin}[\theta]$

Riemann tensor

The Riemann tensor is given by means of the Christoffel symbols $R^{\mu}{}_{\nu\rho\sigma} = \Gamma^{\mu}_{\nu\sigma,\rho} - \Gamma^{\mu}_{\nu\rho,\sigma} + \Gamma^{\mu}_{\rho\lambda}\Gamma^{\lambda}_{\nu\sigma} - \Gamma^{\mu}_{\sigma\lambda}\Gamma^{\lambda}_{\nu\rho}$

```
riemann := riemann = Table[D[affine[[ν, σ, μ]], coord[[ρ]]] - D[affine[[ν, ρ, μ]], coord[[σ]]] +
     Sum[affine[[ρ, λ, μ]] affine[[ν, σ, λ]] - affine[[σ, λ, μ]] affine[[ν, ρ, λ]], {λ, 1, n}],
    {μ, 1, n}, {ν, 1, n}, {ρ, 1, n}, {σ, 1, n}]
```

The Riemann tensor with lower indices reads $R_{\mu\nu\rho\sigma} = g_{\mu\kappa}\, R^{\kappa}{}_{\nu\rho\sigma}$

```
riemannDn := riemannDn = Table[Simplify[Sum[metric[[μ, κ]] riemann[[κ, ν, ρ, σ]], {κ, 1, n}]],
    {μ, 1, n}, {ν, 1, n}, {ρ, 1, n}, {σ, 1, n}]

listRiemann := Table[If[UnsameQ[riemannDn[[μ, ν, ρ, σ]], 0],
    {Style[Subscript[R, Row[{coord[[μ]], coord[[ν]], coord[[ρ]], coord[[σ]]}]], 16],
     "=", riemannDn[[μ, ν, ρ, σ]]}], {ν, 1, n}, {μ, 1, ν}, {σ, 1, n}, {ρ, 1, σ}]

TableForm[Partition[DeleteCases[Flatten[listRiemann], Null], 3], TableSpacing → {2, 2}]
```

$R_{\theta\phi\theta\phi} = -k^2 \left(-1 + k^2\right) r^2\, \mathrm{Sin}[\theta]^2$

Ricci tensor

The Ricci tensor follows from the contraction of the Riemann tensor: $R_{\mu\nu} = R^{\rho}{}_{\mu\rho\nu}$

```
ricci := ricci = Table[Simplify[Sum[riemann[[ρ, μ, ρ, ν]], {ρ, 1, n}]], {μ, 1, n}, {ν, 1, n}]

listRicci :=
 Table[If[UnsameQ[ricci[[μ, ν]], 0], {Style[Subscript[R, Row[{coord[[μ]], coord[[ν]]}]], 16],
     "=", Style[ricci[[μ, ν]], 16]}], {ν, 1, 4}, {μ, 1, ν}]

TableForm[Partition[DeleteCases[Flatten[listRicci], Null], 3], TableSpacing → {1, 2}]
```

$R_{\theta\theta} = 1 - k^2$

$R_{\phi\phi} = -\left(-1 + k^2\right) \mathrm{Sin}[\theta]^2$

Ricci scalar

The Ricci scalar is the given by the contraction of the Ricci tensor $R = R^{\alpha}{}_{\alpha}$

```
ricciscalar :=
 ricciscalar = Simplify[Sum[Sum[inversemetric[[μ, ν]] ricci[[ν, μ]], {μ, 1, n}], {ν, 1, n}]]
```

```
ricciscalar
```

$$\frac{2 - 2\,k^2}{k^2\,r^2}$$

Kretschman scalar

The Kretschman scalar is defined as $K = R_{\alpha\beta\gamma\delta}\, R^{\alpha\beta\gamma\delta} = R^{\alpha\beta}{}_{\gamma\delta} R^{\gamma\delta}{}_{\alpha\beta}$

```
riemannUp :=
 riemannUp = Table[Simplify[Sum[inversemetric[[ν, κ]] riemann[[μ, κ, ρ, σ]], {κ, 1, n}]],
   {μ, 1, n}, {ν, 1, n}, {ρ, 1, n}, {σ, 1, n}]
```

```
kretschman := Simplify[
  Sum[Sum[Sum[Sum[riemannUp[[μ, ν, ρ, σ]] riemannUp[[ρ, σ, μ, ν]], {μ, 1, n}], {ν, 1, n}],
    {ρ, 1, n}], {σ, 1, n}]]
```

```
kretschman
```

$$\frac{4\,\left(-1 + k^2\right)^2}{k^4\,r^4}$$